

Počítačový algebraický systém Maple jako pomůcka při studiu předmětu Matematika I a II.

Pavel Pokorný
Praha, 1. února 2004

Diferenciální rovnice

Obsah

1	Matematické příklady	1
1.1	Úvod	1
1.2	Obecné řešení homogenní lineární diferenciální rovnice	1
1.3	Partikulární řešení homogenní lineární diferenciální rovnice . .	2
1.4	Obecné řešení nehomogenní lineární diferenciální rovnice . . .	4
1.5	Partikulární řešení nehomogenní lineární diferenciální rovnice .	4
1.6	Obecné řešení autonomní nelineární diferenciální rovnice . . .	5
1.7	Partikulární řešení autonomní nelineární diferenciální rovnice .	5
1.8	Obecné řešení lineární diferenciální rovnice 2. řádu	6
1.9	Partikulární řešení lineární diferenciální rovnice 2. řádu s po- čátečními podmínkami	6
1.10	Partikulární řešení lineární diferenciální rovnice 2. řádu s okra- jovými podmínkami	7
1.11	Obecné řešení soustavy lineárních diferenciálních rovnic	7
1.12	Partikulární řešení soustavy lineárních diferenciálních rovnic .	8
1.13	Numerické řešení a řešení ve tvaru řady	8
2	Technické aplikace	10
2.1	Příklad mechanický	10
2.2	Příklad elektrický	13
2.3	Příklad chemický	14
2.3.1	Reakce za konstantního objemu	14
2.3.2	Reakce za konstantního tlaku	15

1 Matematické příklady

1.1 Úvod

Program Maple umožňuje hledat řešení diferenciálních rovnic pomocí příkazu `dsolve`. Ukážeme si jeho použití na jednoduchých příkladech pro lineární i nelineární rovnice, pro hledání obecného i partikulárního řešení, pro jednu rovnici i pro soustavu rovnic, pro symbolické i pro numerické řešení. Na závěr se podíváme na aplikace diferenciálních rovnic při studiu mechanických, elektrických a chemických dynamických systémů.

1.2 Obecné řešení homogenní lineární diferenciální rovnice

Příklad 1.

Nalezněte obecné řešení homogenní lineární diferenciální rovnice

$$y' = -3y.$$

Postup:

Příkaz `dsolve` zavoláme se dvěma argumenty uzavřenými v kulatých závorkách a oddělených čárkou. První argument je diferenciální rovnice, kterou chceme řešit, druhý argument je neznámá funkce. Diferenciální rovnice obsahuje derivaci neznámé funkce, kterou zadáme příkazem `diff`. Ukončíme-li příkaz středníkem, potom se vypíše výsledek na display. Ukončíme-li jej dvojtečkou, potom se výsledek nevypisuje.

```
> dsolve( diff(y(x),x) = -3 * y(x), y(x));  
y(x) = _C1 e(-3 x)
```

Toto obecné řešení obsahuje jednu integrační konstantu, zde označenou jako `_C1`.

1.3 Partikulární řešení homogenní lineární diferenciální rovnice

Příklad 2.

Nalezněte partikulární řešení homogenní lineární diferenciální rovnice

$$y' = -3y$$

splňující počáteční podmínku

$$y(0) = 27.$$

Postup:

Obě rovnice, tedy diferenciální rovnici a počáteční podmínku, uzavřeme do složených nebo do hranatých závorek a oddělíme čárkou. Hranaté závorky se používají pro označení seznamu prvků (ang. list), tedy uspořádané n-tice prvků; složené závorky se používají pro označení množiny (ang. set), tedy skupiny prvků, kde nezáleží na pořadí. V našem případě lze použít kterýkoliv z těchto dvou způsobů.

```
> dsolve({diff(y(x),x)=-3*y(x),y(0)=27},y(x));
```

$$y(x) = 27 e^{(-3x)}$$

Budeme-li s řešením dále pracovat, uložíme si je do proměnné, kterou si pojmenujeme např. **reseni**.

```
> reseni:=dsolve({diff(y(x),x)=-3*y(x),y(0)=27},y(x));
```

$$reseni := y(x) = 27 e^{(-3x)}$$

Vidíme, že řešení je vlastně ve tvaru rovnice

$$y(x) = 27 e^{-3x}.$$

Pravou stranu (ang. right-hand-side) této rovnice, tedy samotné řešení, lze získat příkazem **rhs**

```
> rhs(reseni);
```

$$27 e^{(-3x)}$$

Chceme-li vyhodnotit (ang. evaluate) hodnotu řešení např. pro $x = 3$ použijeme příkaz

```
> eval(rhs(reseni),x=3);
```

$$27 e^{(-9)}$$


Zajímá-li nás přibližná hodnota v desetinném tvaru, zadáme místo přesné hodnoty $x = 3$ přibližnou hodnotu $x = 3.0$ (nulu na konci lze vynechat, ale desetinnou tečku nikoliv)

```
> eval(rhs(reseni),x=3.);
```

$$0.003332064711$$

A nakonec si ukážeme, jak lze nakreslit graf řešení diferenciální rovnice pro zadaný interval hodnot nezávisle proměnné

```
> plot(rhs(reseni),x=0..1);
```



Existuje ještě druhý způsob, jak pracovat s řešením diferenciální rovnice, než použití příkazu `rhs`. Můžeme přiřadit (ang. `assign`) pravou stranu rovnice levé straně rovnice příkazem `assign`. Tedy té rovnice, kterou nám vrátí příkaz `dsolve`, nikoliv rovnice, kterou chceme vyřešit. Celý výpočet potom může vypadat takto

```
> reseni:=dsolve({diff(y(x),x)=-3*y(x),y(0)=27},y(x));
                         $reseni := y(x) = 27 e^{(-3x)}$ 
> assign(reseni);
> y(x);
                         $27 e^{(-3x)}$ 
> eval(y(x),x=3);
                         $27 e^{(-9)}$ 
> eval(y(x),x=3.);
                        0.003332064711
```

1.4 Obecné řešení nehomogenní lineární diferenciální rovnice

Příklad 3.

Nalezněte obecné řešení nehomogenní lineární diferenciální rovnice

$$y' = -3y + x$$

Postup:

```
> dsolve(diff(y(x),x)=-3*y(x)+x, y(x));
```

$$y(x) = \frac{x}{3} - \frac{1}{9} + C_1 e^{(-3x)}$$

1.5 Partikulární řešení nehomogenní lineární diferenciální rovnice

Příklad 4.

Nalezněte partikulární řešení nehomogenní lineární diferenciální rovnice

$$y' = -3y + x$$

vyhovující počáteční podmínce

$$y(1) = 2$$

Postup:

```
> dsolve({diff(y(x),x)=-3*y(x)+x,y(1)=2},y(x));
```

$$y(x) = \frac{x}{3} - \frac{1}{9} + \frac{16}{9} \frac{e^{(-3x)}}{e^{(-3)}}$$

Tento výsledek lze zjednodušit (ang. simplify)

```
> simplify(%);
```

$$y(x) = \frac{x}{3} - \frac{1}{9} + \frac{16}{9} e^{(-3x+3)}$$

(znak procenta znamená poslední výsledek).

1.6 Obecné řešení autonomní nelineární diferenciální rovnice

Příklad 5.

Nalezněte obecné řešení autonomní nelineární diferenciální rovnice

$$y' = y^2$$

Postup:

> `dsolve(diff(y(x),x)=y(x)^2, y(x));`

$$y(x) = -\frac{1}{x - C1}$$

1.7 Partikulární řešení autonomní nelineární diferenciální rovnice

Příklad 6.

Nalezněte partikulární řešení autonomní nelineární diferenciální rovnice

$$y' = y^2$$

vyhovující počáteční podmínce

$$y(2) = 1.$$

Postup:

> `dsolve({diff(y(x),x)=y(x)^2, y(2)=1}, y(x));`

$$y(x) = -\frac{1}{x - 3}$$

1.8 Obecné řešení lineární diferenciální rovnice 2. řádu

Příklad 7.

Nalezněte obecné řešení lineární diferenciální rovnice 2. řádu

$$y'' + 11y' + 30y = 0.$$

Postup:

Pro zadání druhé derivace neznámé funkce použijeme příkaz `diff(y(x), x$2)`

```
> dsolve(diff(y(x), x$2)+11*diff(y(x), x)+30*y(x), y(x));  
y(x) = _C1 e(-6 x) + _C2 e(-5 x)
```

1.9 Partikulární řešení lineární diferenciální rovnice 2. řádu s počátečními podmínkami

Příklad 8.

Nalezněte partikulární řešení lineární diferenciální rovnice 2. řádu

$$y'' + 11y' + 30y = 0$$

vyhovující počátečním podmínkám

$$y(0) = 2$$

a

$$y'(0) = 4.$$

Postup:

Pro zadání derivace neznámé funkce v počáteční podmínce použijeme výraz `D(y)(0)`. Je-li rovnic více, můžeme je zadávat postupně, ukládat do proměnných (zde `rovnice` a `pocpodm`) a v příkazu `dsolve` se na ně odkázat jménem proměnné, do které jsme si je uložili.

```
> rovnice:=diff(y(x), x$2)+11*diff(y(x), x)+30*y(x);  
rovnice := ( $\frac{d^2}{dx^2} y(x)$ ) + 11 ( $\frac{d}{dx} y(x)$ ) + 30 y(x)  
> pocpodm:=y(0)=2,D(y)(0)=4;  
pocpodm := y(0) = 2, D(y)(0) = 4  
> dsolve({rovnice,pocpodm}, y(x));  
y(x) = -14 e(-6 x) + 16 e(-5 x)
```

1.10 Partikulární řešení lineární diferenciální rovnice 2. řádu s okrajovými podmínkami

Příklad 9.

Nalezněte partikulární řešení lineární diferenciální rovnice z předchozího pří-

kladu vyhovující okrajovým podmínkám

$$y(0) = 2$$

a

$$y(1) = 7.$$

Postup:

```
> okrpodm:=y(0)=2,y(1)=7;  
      okrpodm := y(0) = 2, y(1) = 7  
> dsolve({rovnice,okrpodm}, y(x));  
      
$$y(x) = -\frac{(-7 + 2e^{(-6)})e^{(-5x)}}{e^{(-5)} - e^{(-6)}} + \frac{(2e^{(-5)} - 7)e^{(-6x)}}{e^{(-5)} - e^{(-6)}}$$

```

1.11 Obecné řešení soustavy lineárních diferenciálních rovnic

Příklad 10.

Nalezněte obecné řešení soustavy diferenciálních rovnic

$$x' = x + 3y$$

a

$$y' = -4x - 6y.$$

Postup:

```
> rovnice1:=diff(x(t),t)=x(t) + 3 * y(t);  
      rovnice1 :=  $\frac{d}{dt}x(t) = x(t) + 3y(t)$   
> rovnice2:=diff(y(t),t)= -4 * x(t) - 6 * y(t);  
      rovnice2 :=  $\frac{d}{dt}y(t) = -4x(t) - 6y(t)$   
> dsolve({rovnice1,rovnice2},{x(t),y(t)});  
      
$$\{y(t) = -_C1 e^{(-2t)} - \frac{4}{3}_C2 e^{(-3t)}, x(t) = _C1 e^{(-2t)} + _C2 e^{(-3t)}\}$$

```


1.12 Partikulární řešení soustavy lineárních diferenciálních rovnic

Příklad 11.

Nalezněte partikulární řešení soustavy diferenciálních rovnic z předchozího příkladu vyhovující počátečním podmínkám

$$x(0) = 2$$

a

$$y(0) = 3.$$

Postup:

```
> pocpodm1:=x(0)=2;
      pocpodm1 := x(0) = 2
> pocpodm2:=y(0)=3;
      pocpodm2 := y(0) = 3
> dsolve({rovnice1,rovnice2,pocpodm1,pocpodm2}, {x(t),y(t)});
      {y(t) = -17 e(-2 t) + 20 e(-3 t), x(t) = 17 e(-2 t) - 15 e(-3 t)}
```

1.13 Numerické řešení a řešení ve tvaru řady

Některé diferenciální rovnice (chce se mi říci většina) mají řešení obsahující speciální matematické funkce (např. gamma funkci Γ , Besselovu funkci apod.) nebo dokonce řešení, které nelze zapsat výrazem konečné délky. Takové rovnice lze řešit pomocí mocninných řad nebo přibližně numericky.

Příklad 12.

Nalezněte partikulární řešení diferenciální rovnice

$$y' = x^2 + y^2$$

vyhovující počáteční podmínce

$$y(0) = 3.$$

Postup:

```
> orisek:={diff(y(x),x)=x^2+y(x)^2,y(0)=3};
      orisek := {y(0) = 3,  $\frac{d}{dx} y(x) = x^2 + y(x)^2$ }
> dsolve(orisek,y(x));
```

$$y(x) = -\frac{x \left(-\frac{(\Gamma(\frac{3}{4})^2 - 3\pi) \text{BesselJ}(\frac{-3}{4}, \frac{x^2}{2})}{\Gamma(\frac{3}{4})^2} + \text{BesselY}(\frac{-3}{4}, \frac{x^2}{2}) \right)}{-\frac{(\Gamma(\frac{3}{4})^2 - 3\pi) \text{BesselJ}(\frac{1}{4}, \frac{x^2}{2})}{\Gamma(\frac{3}{4})^2} + \text{BesselY}(\frac{1}{4}, \frac{x^2}{2})}$$

Příklad 13.

Nalezněte řešení z předchozího příkladu pomocí mocninné řady.

Postup:

```
> dsolve(orisek,y(x),series);
```


$$y(x) = 3 + 9x + 27x^2 + \frac{244}{3}x^3 + \frac{487}{2}x^4 + \frac{3654}{5}x^5 + O(x^6)$$

Příklad 14.

Nalezněte přibližné numerické řešení předchozího příkladu pro $x = 0.2$ a nakreslete graf řešení pro $x \in (0, 0.3)$.

Postup:

```
> numerickereseni:=dsolve(orisek,y(x),numeric);
      numerickereseni := proc(x_rkf45) ... end proc
> numerickereseni(0.2);
      [x = 0.2, y(x) = 7.50526519376205314]
> plots[odeplot](numerickereseni,0..0.3);
```



2 Technické aplikace

Diferenciální rovnice patří mezi nejužitečnější nástroje, které dává matematika přírodním a technickým vědám pro popis dynamických systémů. Ukážeme si tři příklady, a to jeden mechanický, jeden elektrický a jeden chemický.


2.1 Příklad mechanický

Určete dobu T , za kterou dopadne na zem člověk, kterého budeme považovat za hmotný bod o hmotnosti m , stojící uprostřed žebříku délky $L = 40$ m. Žebřík stojí na vodorovné hladké podlaze a je opřen o svislou hladkou stěnu pod úhlem 45° . V čase $t = 0$ se žebřík uvede do pohybu tak, že jeho dolní konec klouže po hladké podlaze a jeho horní konec klouže po hladké stěně (viz. obrázek). Hmotnost žebříku zanedbejte a uvažujte tíhové zrychlení $g = 10 \text{ m s}^{-2}$.

Řešení:

Z obrázku určíme souřadnice osoby stojící uprostřed žebříku

$$x = \frac{L}{2} \sin \alpha$$



Obr. 1: Žebřík stojící na hladké podlaze opřený o hladkou stěnu

$$y = \frac{L}{2} \cos \alpha.$$

To je vlastně parametricky zadaná křivka v rovině $x - y$ s parametrem α , který závisí na čase t .

Protože

$$x^2 + y^2 = \left(\frac{L}{2} \sin \alpha\right)^2 + \left(\frac{L}{2} \cos \alpha\right)^2 = \left(\frac{L}{2}\right)^2,$$

má střed žebříku konstantní vzdálenost od počátku – rohu mezi podlahou a stěnou, tedy padající osoba se pohybuje po kružnici, jako by klouzala např. po hladké kouli.

Obě složky rychlosti pohybu jsou

$$v_x = \frac{dx}{dt} = \frac{L}{2} \cos \alpha \frac{d\alpha}{dt}$$

$$v_y = \frac{dy}{dt} = -\frac{L}{2} \sin \alpha \frac{d\alpha}{dt}.$$

Rychlost pohybu osoby je potom

$$v = \sqrt{v_x^2 + v_y^2} = \frac{L}{2} \frac{d\alpha}{dt}.$$

Ze zákona zachování energie

$$\frac{1}{2}mv^2 + mgy = mgy_0$$

sestavíme diferenciální rovnici popisující časový vývoj úhlu α

$$\frac{d\alpha}{dt} = \frac{2}{L}v = \frac{2}{L}\sqrt{2g(y_0 - y)} = \frac{2}{L}\sqrt{2g\frac{L}{2}(\cos\alpha_0 - \cos\alpha)} = \sqrt{\frac{4g}{L}(\cos\alpha_0 - \cos\alpha)}.$$

A pro zadané hodnoty

$$\frac{d\alpha}{dt} = \sqrt{\cos\frac{\pi}{4} - \cos\alpha}.$$

My ovšem tuto rovnici řešit nebudeme. Pouze najdeme

$$dt = \frac{d\alpha}{\sqrt{\cos\frac{\pi}{4} - \cos\alpha}}$$

a integrací určíme dobu T , za kterou nešťastník dopadne na zem

$$T = \int_0^T dt = \int_{\frac{\pi}{4}}^{\frac{\pi}{2}} \frac{d\alpha}{\sqrt{\cos\frac{\pi}{4} - \cos\alpha}}.$$

To lze spočítat v systému Maple následujícími příkazy

```
> int(1/sqrt(cos(Pi/4)-cos(a)),a=Pi/4..Pi/2);
```

$$\sqrt{2} \operatorname{EllipticF}\left(\frac{2^{(3/4)}}{\sqrt{2+\sqrt{2}}}, \frac{\sqrt{2+\sqrt{2}}}{2}\right) - \sqrt{2} \operatorname{EllipticF}\left(\frac{\sqrt{4\sin(\frac{\pi}{8})^2 - 2 + \sqrt{2}}}{\sqrt{2+\sqrt{2}}\sin(\frac{\pi}{8})}, \frac{\sqrt{2+\sqrt{2}}}{2}\right)$$

```
> simplify(%);
```

$$\sqrt{2} \operatorname{EllipticF}\left(\frac{2^{(3/4)}}{\sqrt{2+\sqrt{2}}}, \frac{\sqrt{2+\sqrt{2}}}{2}\right)$$


```
> evalf(%);
```

2.007220538

Závěr: osoba dopadne na zem přibližně za $T = 2$ s.

2.2 Příklad elektrický

Kondenzátor o kapacitě $C = 10 \mu\text{F}$ a nabitý na napětí $U_0 = 30 \text{ V}$ se vybíjí přes odpor $R = 5 \text{ M}\Omega$. Určete napětí kondenzátoru za jednu minutu.



Obr. 2: Schema elektrického obvodu s kondenzátorem a odporem

Postup:

Časová změna napětí kondenzátoru je úměrná proudu

$$C \frac{dU}{dt} = -I.$$

Záporné znaménko odpovídá skutečnosti, že vytékající proud kondenzátor vybíjí. Proud I je z Ohmova zákona roven poměru napětí U a odporu R takže

$$C \frac{dU}{dt} = -\frac{U}{R},$$

což dává lineární autonomní diferenciální rovnici prvního řádu

$$\frac{dU}{dt} = -\frac{U}{RC}.$$

Tu lze řešit separací proměnných, pomocí charakteristické rovnice nebo na počítači pomocí systému Maple příkazem

```
> dsolve({diff(u(t),t)=-u(t)/(R*C),u(0)=u0},u(t));
```

$$u(t) = u0 e^{(-\frac{t}{RC})}$$

```
> rhs(%);
```

$$u0 e^{(-\frac{t}{RC})}$$


```
> eval(%, [R=5e6,C=10e-6,u0=30,t=60]);
```

$$9.035826357$$

Závěr: napětí kondenzátoru za jednu minutu klesne z 30 V na přibližně 9 V.

2.3 Příklad chemický

Určete koncentraci NO_2 v čase $t = 0.8$ s, jestliže vzniká reakcí



a počáteční koncentrace NO v čase $t = 0$ je $c_{\text{NO}}(0) = 2 \text{ mol/m}^3$, počáteční koncentrace O_2 je $c_{\text{O}_2}(0) = 3 \text{ mol/m}^3$ a počáteční koncentrace NO_2 je nulová. Rychlostní konstanta této reakce při teplotě 30°C je $k = 0.01325 \text{ m}^6\text{mol}^{-2}\text{s}^{-1}$. Uvažujte, že reakce probíhá (a) při konstantním objemu, (b) při konstantním tlaku.

Postup:

Z chemické rovnice vidíme, že rychlost r této reakce (definovaná jako časová změna látkového množství – tj. počtu molů – v jednotce objemu dělená stechiometrickým koeficientem) je úměrná druhé mocnině koncentrace NO a první mocnině koncentrace O_2 s konstantou úměrnosti k , tedy

$$r = k c_{\text{NO}}^2 c_{\text{O}_2}.$$

Protože s každým vzniklým molem NO_2 ubude jeden mol NO a také půl molu O_2 , je okamžité látkové množství NO v čase t rovno

$$n_{\text{NO}}(t) = n_{\text{NO}}(0) - n_{\text{NO}_2}(t)$$

a okamžité látkové množství O_2 v čase t je rovno

$$n_{\text{O}_2}(t) = n_{\text{O}_2}(0) - \frac{1}{2}n_{\text{NO}_2}(t).$$

2.3.1 Reakce za konstantního objemu

Je-li plyn uzavřen v neprodyšné nádobě o konstantním objemu V , budou koncentrace

$$c = \frac{n}{V}$$

úměrné látkovým množstvím n a rychlost reakce bude

$$r = k (c_{\text{NO}}(0) - c_{\text{NO}_2}(t))^2 (c_{\text{O}_2}(0) - \frac{1}{2}c_{\text{NO}_2}(t)).$$

Růst koncentrace NO_2 lze popsat diferenciální rovnicí

$$\frac{dc_{\text{NO}_2}(t)}{dt} = 2 k (c_{\text{NO}}(0) - c_{\text{NO}_2}(t))^2 (c_{\text{O}_2}(0) - \frac{1}{2}c_{\text{NO}_2}(t)).$$

Označíme-li si, pro zkrácení zápisu, $a = c_{NO}(0)$, $b = c_{O_2}(0)$, a $y = c_{NO_2}(t)$, dostaneme

$$y' = k(a - y)^2(2b - y).$$

Nyní již můžeme použít systém Maple pro určení $y(0.8)$

```
> k := 0.01325: a := 2: b := 3:
> rovnice:=diff(y(t),t)=k*(a-y(t))^2*(2*b-y(t)):
> dsolve({rovnice,y(0)=0},y(t),numeric)(0.8);
[t = 0.8, y(t) = 0.221814327883515211]
```

Závěr: při reakci za konstantního objemu bude v čase $t = 0.8$ s koncentrace NO_2 přibližně rovna 0.22 mol/m^3 .

Reakcí se mění látkové množství (počet molů plynu), protože dva moly NO a jeden mol O_2 dají vzniknout dvěma molům NO_2 . Takže při konstantním objemu klesá tlak. Tento pokles tlaku by bylo možné vyrovnávat tak, že bychom připouštěli do objemu inertní plyn, např. N_2 . Potom by ale částečně unikaly i plyny účastníci se reakce. Zanedbáme-li únik plynu, zůstává výše uvedený výpočet v platnosti.

2.3.2 Reakce za konstantního tlaku

Probíhá-li reakce za konstantního tlaku p , bude se objem V v důsledku klesajícího látkového množství zmenšovat. V tomto případě je výhodnější uvažovat látkové množství n místo koncentrace c

$$\frac{dn_{NO_2}(t)}{dt} = 2 V(t) k \left(\frac{n_{NO}(0) - n_{NO_2}(t)}{V(t)} \right)^2 \left(\frac{n_{O_2}(0) - \frac{1}{2}n_{NO_2}(t)}{V(t)} \right).$$

Okamžitý objem $V(t)$ je

$$V(t) = V(0) - \frac{1}{2} V_1 n_{NO_2}(t),$$

kde $V_1 = \frac{RT}{p} \doteq 0.0246 \text{ m}^3/\text{mol}$ je objem jednoho molu plynu při daném tlaku p a teplotě T .

Potom diferenciální rovnice popisující časový vývoj látkového množství NO_2 je

$$\frac{dn_{NO_2}(t)}{dt} = 2k \frac{(n_{NO}(0) - n_{NO_2}(t))^2 (n_{O_2}(0) - \frac{1}{2}n_{NO_2}(t))}{(V(0) - \frac{1}{2} V_1 n_{NO_2}(t))^2}.$$

Poté, co vyřešíme, např. numericky, tuto diferenciální rovnici, můžeme určit koncentraci NO_2 v čase t

$$c_{\text{NO}_2}(t) = \frac{n_{\text{NO}_2}(t)}{V(t)} = \frac{n_{\text{NO}_2}(t)}{V(0) - \frac{1}{2} V_1 n_{\text{NO}_2}(t)}.$$

Pro přepočet počátečních koncentrací na počáteční látková množství můžeme uvažovat libovolný, např. jednotkový, počáteční objem $V(0) = 1 \text{ m}^3$.

Označíme-li si, pro zkrácení zápisu, $a = n_{\text{NO}}(0) = c_{\text{NO}}(0) V(0)$, $b = n_{\text{O}_2}(0) = c_{\text{O}_2}(0) V(0)$, $y = n_{\text{NO}_2}(t)$ a zvolíme-li $V(0) = 1$, dostaneme

$$y' = k \frac{(a - y)^2 (2b - y)}{(1 - V_1 y/2)^2}$$

a

$$c_{\text{NO}_2}(t) = \frac{y}{1 - V_1 y/2}.$$

Nyní již můžeme použít systém Maple pro určení $c_{\text{NO}_2}(0.8)$

```
> k:=0.01325: a:=2: b:=3: v1:=0.0246:
> rovnice2:=diff(y(t),t)=k*(a-y(t))^2*(2*b-y(t))/(1-v1*y(t)/2)^2;
rovnice2 :=  $\frac{d}{dt} y(t) = \frac{0.01325 (2 - y(t))^2 (6 - y(t))}{(1 - 0.01230000000 y(t))^2}$ 
> reseni:=dsolve({rovnice2,y(0)=0},y(t),numeric)(0.8);
reseni := [t = 0.8, y(t) = 0.222369027775583428]
> nNO2:=rhs(reseni[2]);
nNO2 := 0.222369027775583428
> cNO2:=nNO2/(1-v1*nNO2/2);
cNO2 := 0.2229789061
```

Závěr: při reakci za konstantního tlaku bude v čase $t = 0.8 \text{ s}$ koncentrace NO_2 přibližně rovna 0.22298 mol/m^3 , tedy přibližně o 0.5 % více než při reakci za konstantního objemu. Tento rozdíl je malý, protože počáteční koncentrace reagujících plynů jsou nízké. Přesněji

$$c_{\text{NO}}(0) + c_{\text{O}_2}(0) \ll \frac{1}{V_1} \doteq \frac{1}{0.0246} \doteq 41 \text{ mol/m}^3,$$

takže většina objemu je zaplněna jinými plyny, jejichž reakce neuvažujeme.